
Subject: Tree widget woes

Posted by [robert.dimeo](#) on Fri, 12 Dec 2003 16:37:47 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hi,

I work quite a bit with tree widgets in my IDL code and I have run into a problem. I use tree widgets to represent data in a folder hierarchy and clicking on the leaves usually results in a plot of a single data set or a multiple data set depending on whether or not the user selects more than one leaf. I want to be able to create dynamically the same kind of folder layout at the root level. If the multiple keyword is set in the tree widget then I have problems with where the duplicate folder layout gets realized if one of the leaf widgets is selected. I tried unselecting the selected leaf widget prior to adding the folder hierarchy but this does not solve the problem.

I have found that this problem does not occur on LINUX but only on WINDOWS. (might be a useful hint there...?)

If the description above does not make sense I hope the code below will help...

The code at the end of this message illustrates the problems I'm having. To run it, compile it and type TREE_WIDGET_TEST,MULT = 1. The keyword MULT simply tells the tree widget to allow multiple leaf/folder selections. Select one of the leaves (labeled DATA) and then press the button labeled ADD_FOLDER. What I want to happen is for a whole new hierarchy of folders to appear at the root level. It actually appears under the leaf that was selected. Now if you run the program with the keyword MULT=0 you get the desired behavior. Unfortunately I want to be able to select multiple files AND be able to put in the new folder hierarchy. Can I have my cake and eat it too? Any help would be greatly appreciated.

Thanks,

Rob

```
.....  
!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!  
pro tt_add_folder,event  
file_tree = widget_info(event.top,find_by_uname = 'FILE_TREE')  
group = widget_tree(file_tree,value = 'Data Group',/folder,$  
    /expanded)  
leaf1 = widget_tree(group, value = 'Data')  
leaf2 = widget_tree(group, value = 'Data')  
end
```

```

.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tt_event,event
uname = widget_info(event.id,/uname)
case uname of
'QUIT': widget_control,event.top,/destroy
'ADD_FOLDER': tt_add_folder,event
else:
endcase
end
.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tree_widget_test,mult = mult
if n_elements(mult) eq 0 then mult = 0
register_name = 'tw_test'
tlb = widget_base(title = 'Tree Widget Test',/col,/tlb_frame_attr)
tree_base = widget_base(tlb,/col)
file_tree = widget_tree(tree_base,multiple = mult,uname = 'FILE_TREE')

group = widget_tree(file_tree,value = 'Data Group',/folder,$
    /expanded)
leaf1 = widget_tree(group, value = 'Data')
leaf2 = widget_tree(group, value = 'Data')
void = widget_button(tlb,value = 'ADD FOLDER',uname = 'ADD_FOLDER')
void = widget_button(tlb,value = 'QUIT',uname = 'QUIT')
widget_control,tlb,/realize
xmanager,register_name,tlb,/no_block,event_handler = 'tt_event'
end

```

Subject: Re: Tree widget woes

Posted by [robert.dimeo](#) on Mon, 15 Dec 2003 14:49:22 GMT

[View Forum Message](#) <> [Reply to Message](#)

robert.dimeo@nist.gov (Rob Dimeo) wrote in message
news:<cb539436.0312120837.5f2aa843@posting.google.com>...

> then press the button labeled ADD_FOLDER. What I want to happen is
> for a whole new hierarchy of folders to appear at the root level. It
> actually appears under the leaf that was selected. Now if you run the
> program with the keyword MULT=0 you get the desired behavior.
> Unfortunately I want to be able to select multiple files AND be able
> to put in the new folder hierarchy. Can I have my cake and eat it
> too? Any help would be greatly appreciated.
>

So I guess I'll reply to my own posting with a workaround (the silence
was deafening...anyone else out there using the tree widgets?). The
code below shows how I implemented the workaround. The idea is to add
a folder (node) at the root level and make all additional folders

children of that one. I would rather not have to put in the root folder but it's the only fix that I could come up with.

```
.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tt_add_folder,event
; Get the widget id of the root node...
root = widget_info(event.top,find_by_undef = 'ROOT')
; and make the additional folders and leaves children of the root.
group = widget_tree(root,value = 'Data Group',/folder,$
    /expanded)
leaf1 = widget_tree(group, value = 'Data')
leaf2 = widget_tree(group, value = 'Data')
end
.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tt_event,event
undef = widget_info(event.id,/undef)
case undef of
'QUIT': widget_control,event.top,/destroy
'ADD_FOLDER': tt_add_folder,event
else:
endcase
end
.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tree_widget_test,mult = mult
if n_elements(mult) eq 0 then mult = 0
register_name = 'tw_test'
tlb = widget_base(title = 'Tree Widget Test',/col,/tlb_frame_attr)
tree_base = widget_base(tlb,/col)
file_tree = widget_tree(tree_base,multiple = mult,undef = 'FILE_TREE')
; Define the root node...
root = widget_tree(file_tree,value = 'Root',/folder,/expanded, $
    undef = 'ROOT')
; and make all the rest children of that node
group = widget_tree(root,value = 'Data Group',/folder,$
    /expanded)
leaf1 = widget_tree(group, value = 'Data')
leaf2 = widget_tree(group, value = 'Data')
void = widget_button(tlb,value = 'ADD_FOLDER',undef = 'ADD_FOLDER')
void = widget_button(tlb,value = 'QUIT',undef = 'QUIT')
widget_control,tlb,/realize
xmanager,register_name,tlb,/no_block,event_handler = 'tt_event'
end
```

Subject: Re: Tree widget woes

Posted by [robert.dimeo](#) on Thu, 18 Dec 2003 13:13:16 GMT

[View Forum Message](#) <> [Reply to Message](#)

robert.dimeo@nist.gov (Rob Dimeo) wrote in message
news:<cb539436.0312150649.442558b5@posting.google.com>...
> robert.dimeo@nist.gov (Rob Dimeo) wrote in message
news:<cb539436.0312120837.5f2aa843@posting.google.com>...

>
> So I guess I'll reply to my own posting with a workaround (the silence
> was deafening...anyone else out there using the tree widgets?). The
> code below shows how I implemented the workaround. The idea is to add
> a folder (node) at the root level and make all additional folders
> children of that one. I would rather not have to put in the root
> folder but it's the only fix that I could come up with.

Hi again,

The good folks at RSI have contacted me that this is indeed a bug and
is being investigated.

Rob

Subject: Re: Tree widget woes
Posted by [robert.dimeo](#) on Mon, 22 Dec 2003 21:09:26 GMT
[View Forum Message](#) <> [Reply to Message](#)

robert.dimeo@nist.gov (Rob Dimeo) wrote in message
news:<cb539436.0312180513.60ecac74@posting.google.com>...
> robert.dimeo@nist.gov (Rob Dimeo) wrote in message
news:<cb539436.0312150649.442558b5@posting.google.com>...
>> robert.dimeo@nist.gov (Rob Dimeo) wrote in message
news:<cb539436.0312120837.5f2aa843@posting.google.com>...
>
>>
>> So I guess I'll reply to my own posting with a workaround (the silence
>> was deafening...anyone else out there using the tree widgets?). The
>> code below shows how I implemented the workaround. The idea is to add
>> a folder (node) at the root level and make all additional folders
>> children of that one. I would rather not have to put in the root
>> folder but it's the only fix that I could come up with.
>
> Hi again,
>
> The good folks at RSI have contacted me that this is indeed a bug and
> is being investigated.
>
> Rob

Hi again,

Thanks to the technical support at RSI for pointing out a workaround for the tree widget problem. The workaround is to de-select the entire tree widget prior to adding a new node. The modified code with this workaround is listed below. Also they indicate that this bug is fixed in 6.1.

Rob

```
.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tt_add_folder,event
file_tree = widget_info(event.top,find_by_uname = 'FILE_TREE')
; Here comes the workaround....
widget_control, file_tree, set_tree_select = 0
; That's all there is to it!
group = widget_tree(file_tree,value = 'Data Group',/folder,$
    /expanded)
leaf1 = widget_tree(group, value = 'Data')
leaf2 = widget_tree(group, value = 'Data')
end
.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tt_event,event
uname = widget_info(event.id,/uname)
case uname of
'QUIT': widget_control,event.top,/destroy
'ADD_FOLDER': tt_add_folder,event
else:
endcase
end
.....
,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,
pro tree_widget_test,mult = mult
if n_elements(mult) eq 0 then mult = 0
register_name = 'tw_test'
tlb = widget_base(title = 'Tree Widget Test',/col,/tlb_frame_attr)
tree_base = widget_base(tlb,/col)
file_tree = widget_tree(tree_base,multiple = mult,uname = 'FILE_TREE')

group = widget_tree(file_tree,value = 'Data Group',/folder,$
    /expanded)
leaf1 = widget_tree(group, value = 'Data')
leaf2 = widget_tree(group, value = 'Data')
void = widget_button(tlb,value = 'ADD FOLDER',uname = 'ADD_FOLDER')
void = widget_button(tlb,value = 'QUIT',uname = 'QUIT')
widget_control,tlb,/realize
xmanager,register_name,tlb,/no_block,event_handler = 'tt_event'
end
```
